



NATIONAL RADIO ASTRONOMY OBSERVATORY

ELECTRONICS DIVISION TECHNICAL NOTE NO. 114

TITLE: A REVISED VAX FARANT SPLIT ROUTINE AND RELATED APPLE II
PLOTTING PROGRAM

AUTHOR(S): R. Norrod

DATE: September 27, 1982

DISTRIBUTION: GB Library
CV Library
VLA Library
TU Libraries: Downtown File
Mountain File

J. Payne
H. Hvatum
R. Lacasse
S. Weinreb
P. Napier
M. Balister
C. Burgess
W. Brundage
C. Moore
C. Brockway
J. Coe
G. Behrens
R. Mauzy
R. Norrod
L. D'Addario

A REVISED VAX FARANT SPLOT ROUTINE AND
RELATED APPLE II PLOTTING PROGRAM

R. Norrod

The purpose of this report is to describe some upgrades made to the VAX "SPLOT" routine and the related Apple plotting program described in reference [1]. The purpose of these upgrades is to allow the plotting of S-parameter magnitude and phase versus frequency. In the process of modification, auto-scaling of the data was also added.

The revised FARANT routine is named "SPLOT2" and is called from FARANT with the statement:

```
CALL SPLOT2(I,J,TITLE$,FILENAME$)
```

where I,J are the indexes of the S-parameter to be plotted, TITLE\$ is a literal or string variable containing the plot title, and FILENAME\$ is a literal or string variable containing the filename to which data will be written. The BASIC version of SPLOT2 is resident on the GBELEC VAX directory with the name "SPLOT2.BAS", and a compiled version under the name "SPLOT2.OBJ". A command file, "RUNFAR2", is also provided to link "SPLOT2.OBJ" to the rest of FARANT when \$@RUNFAR2 filename is executed. Here "filename" is a user-edited version of "FARAFT.BAS" containing his circuit description.

Details of transferring "SPLOT" VAX files to the Apple are covered in references [1] and [2] and are not affected by this upgrade. Transferred "SPLOT2" files are plotted by booting the "VAX FARANT PLOTTER" disk and then typing EXEC FARPLOT<C/R>. "FARPLOT" loads various binary utilities needed and then runs the Applesoft program "PLOTTER1.A". The menu of "PLOTTER1.A" is explained below.

G)ET FILE - Loads a data file into memory from disk. The user is prompted for the filename.

- C)LEAR - Erases the graphics screen without affecting data in memory.
- L)OAD PIC - Loads a graphics file onto the HGR screen. Two graphics overlays are provided on the "VAX FARANT PLOTTER" disk: "SMITH", a unit circle Smith chart overlay; and "RECTANGLE", a rectangular overlay for plotting magnitude and phase. Loading either of these overlays sets internal flags that control how data is scaled and plotted.
- A)UTOSCALE - Toggles the automatic scaling function on and off. If "on", the word AUTOSCALE will flash. If "RECTANGLE" is the most recent overlay loaded, then the magnitude plot is scaled to the nearest 10 dB above and below the data minimums and maximums. The phase plot is scaled to ± 180 degrees. If "SMITH" is the most recent overlay loaded, or no overlay has been loaded, then data is scaled for the unit circle.
- P)LOT - Causes the data in memory to be plotted. The type of plot produced is controlled by the most recent overlay loaded. Scaling is controlled by the current status of the autoscale function or by scale values input by the user.
- H)ARDCOPY - Produces a printed copy of the graphics screen.
- T)YPE - Allows the user to type on the graphics screen. Use of this option is explained in reference [1].
- V)IEW - Toggles display between the menu and the graphics screen.
- S)CALE - Accepts inputs for scaling the plots. The type of input requested by the program is controlled by which overlay has most recently been loaded. Current scale values are dis-

S)CALE (Continued):

played and are not changed if the user responds with just a carriage return. If any value is input, then the autoscale function is turned off.

E)XIT - Exits the program to Applesoft.

Compatibility

The "PLOTTER" Apple program described in reference [1] can read files generated by "SPLOT2" although the frequency, magnitude, and phase data will not be read. However, if "FARPLOT" tries to read a data file generated by "SPLOT", it will bomb. Hence, "SPLOT" is not on the GBELEC directory.

Acknowledgments

These programs are modifications of programs written by Stowe Keller. Assistance by Larry D'Addario is appreciated.

References

- [1] Keller, Stowe, "Using the Apple II for Graphics Output from VAX FARANT", NRAO EDTN 113.
- [2] Norrod, Roger, "Use of the VAX-11 FARANT Microwave Design Program from Green Bank", NRAO EDTN 115.

```

10 SUB SPLOT2(I,J,TITLE$,FILENAME$)
20 !WRITES DATA TO FILE SUITABLE FOR DOWNLOADING TO APPLE & PLOTTING
22 !FORMAT-FIELD 1:REAL MIN,REAL MAX,IMAG MIN, IMAG MAX
24 !FIELD 3: NUMBER POINTS
26 !FIELD 4 TO #PTS + 3: REAL,IMAG
27 !REPEAT FOR FREQ, MAGNITUDE, AND PHASE
28 CM$=","
30 COM NOG0%,Z0,F,DBNUM%,DB(101%,18%)
40 IF DB(51%,1%)<>4 THEN CALL PRT(-4,0)
50 C%=2-(I>1)*4-(J>1)*2 !COL. OF DATABASE WITH REAL PART OF S(I,J); NEGATIVE LOG
60 !FIND EXTREMES
65 ALOW=9E30\ RHIGH=-9E30\ ILOW=9E30\ IHIGH=-9E30
70 MLOW=9E30\ MHIGH=-9E30\ ALOW=180\ AHIGH=-180\ FLOW=9E30\ FHIGH=0
75 !
80 FOR R%=1% TO DBNUM%
85 RE=DB(R%,C%)\ IM=DB(R%,C%+1%)\ MG=RE\ AN=IM
90 CALL POLAR(MG,AN) !CALC MAG & ANGLE IN DEGREES
95 IF RE<ALOW THEN ALOW=RE
100 IF RE>RHIGH THEN RHIGH=RE
105 IF IM<ILOW THEN ILOW=IM
110 IF IM>IHIGH THEN IHIGH=IM
115 IF MG<MLOW THEN MLOW=MG
117 IF MG>MHIGH THEN MHIGH=MG
120 IF AN<ALOW THEN ALOW=AN
125 IF AN>AHIGH THEN AHIGH=AN
135 IF DB(R%,1%)<FLOW THEN FLOW=DB(R%,1%)
140 IF DB(R%,1%)>FHIGH THEN FHIGH=DB(R%,1%)
145 NEXT R%
150 MLOW=20*LOG10(MLOW)\ MHIGH=20*LOG10(MHIGH) !CONVERT TO DB
155 OPEN FILENAME$ FOR OUTPUT AS FILE #1, DEFAULTNAME ".DAT"
160 PRINT #1,RLOW;CM$;RHIGH;CM$;ILOW;CM$;IHIGH
162 PRINT #1,DBNUM%
163 FOR R%=1% TO DBNUM%
164 PRINT #1,DB(R%,C%);CM$;DB(R%,C%+1%)
165 NEXT R%
166 PRINT #1,TITLE$
168 PRINT #1,FLOW;CM$;FHIGH;CM$;MLOW;CM$;MHIGH;CM$;ALOW;CM$;AHIGH
169 PRINT #1,DBNUM%
170 FOR R%=1% TO DBNUM%
175 RE=DB(R%,C%)\ IM=DB(R%,C%+1%)\ MG=RE\ AN=IM
180 CALL POLAR(MG,AN)
185 MG=20*LOG10(MG) !CONVERT MAGNITUDE TO DB
190 PRINT #1,DB(R%,1%);CM$;MG;CM$;AN
195 NEXT R%
200 PRINT #1,TITLE$
210 CLOSE #1
220 SUBEND

```

PROGRAM LENGTH= 5389 BYTES VARIABLES= 35 BYTES
 FREE MEMORY= 15075 BYTES
 START=16385 LOMEM=21754 FREE=21789 STRING=36864 HIMEM=36864

```

20 REM PROGRAM * * FARPLOT * * FOR USE WITH FARANT SPLOT2FILES
30 REM MODIFIED BY RDN 19SEPT82
40 REM PLOTTING PROGRAM 1/15/82 FRI
50 DIM A(125,4),CC(3):CC(0) = 2:CC(1) = 6:CC(2) = 1:CC(3) = 5
55 POKE - 16300,0: REM PRIMARY PAGE
57 ONERR GOTO 1000
60 DEF FN HC(X) = CC(2 * X - 4 * INT (.5 * X))
62 DEF FN RC(X) = INT (X + .5)
63 DEF FN RI(X) = INT (10 * X + .5) / 10
65 SCALE= 1: ROT= 0: REM DEFINE PLOTTING ORIENTATION
67 AS% = 1:F% = 0:LA$ = "N":T$ = "T":M% = 0: REM DEFAULTS:AUTOSCALE,
  NO FILE, NO LABELS, TITLE AT TOP, NOT MAG/ANGLE PLOT
70 D$ = CHR$(4):CR$ = CHR$(13)
75 XH = 1.46073298:XL = - XH:YH = 1.0:YL = - YH:UH = XH:UL = XL:UH
  = YH:UL = YL: REM DEFAULT SCALE TO UNIT CIRCLE
77 MH = 0:ML = - 40:PH = 180:PL = - 180:FL = 0:FH = 1: REM DEFAULTS
  FOR MAG/PHASE PLOTS
80 HOME : POKE 35,22: INVERSE : UTAB 23: PRINT "G)ET FILE C)LEAR L)OAD
  PIC ";: IF AS% THEN FLASH
82 PRINT " A)UTOSCALE ": INVERSE
90 UTAB 24: PRINT "P)LOT H)ARDCOPY T)YPE U)IEW S)CALE E)XIT"
100 NORMAL : HOME
110 PRINT "READY ";: GET A$: PRINT A$
120 POKE - 16303,0: REM

130 IF A$ < > "G" THEN 240
140 INPUT "GET FILE: ";A$: IF A$ = "" THEN PRINT D$;"CATALOG": GOTO
  110
147 PRINT D$"OPEN"A$
150 PRINT D$"READ"A$: INPUT XM,XX,YM,YX: REM READ MIN AND MAX VALUES
  OF REAL AND IMAG
155 INPUT NP: REM READ NUMBER OF POINTS
160 FOR A = 0 TO NP - 1
165 INPUT A(A,0),A(A,1)
168 NEXT A ← 170 INPUT TI$
172 INPUT FM,FX,MM,MX,AM,AX: REM READ MIN AND MAX VALUES OF FREQ,MAG,AND
  PHASE
174 INPUT NP
176 FOR A = 0 TO NP - 1
178 INPUT A(A,2),A(A,3),A(A,4)
180 NEXT A

200 INPUT TI$
205 F% = 1: REM FILE LOADED

220 PRINT CHR$(4);"CLOSE"
230 GOTO 110: REM

240 IF A$ < > "A" THEN 300
250 AS% = NOT AS%: REM TOGGLE AUTOSCALE
260 GOTO 80: REM

```

```

300 IF A$ < > "C" THEN 330
310 HGR : POKE - 16303,0
315 U = 0
320 GOTO 80: REM

```

```

330 IF A$ < > "L" THEN 380
340 INPUT "LOAD PIC NAME?";FI$
350 IF NOT LEN (FI$) THEN PRINT D$:"CATALOG": GOTO 110
355 IF FI$ = "SMITH" THEN UH = 1.46073298:UL = - UH:UH = 1.0:UL =
- UH:M% = 0:T$ = "T":LA$ = "N": REM SCALE FOR SMITHCHART
357 IF FI$ = "RECTANGLE" THEN M% = 1:T$ = "B":LA$ = "N"
360 PRINT D$"BLOAD"FI$,"A8192": REM HGR1
365 POKE - 16304,0: POKE - 16297,0: POKE - 16302,0:U = 1
370 GOTO 110: REM

```

```

380 IF A$ < > "S" THEN 445
385 IF M% THEN 424: REM CHECK FOR MAG/ANGLE PLOT
390 PRINT "XLOW /"XL"/ "": INPUT A$: IF A$ < > "" THEN XL = VAL
(A$):AS% = 0
400 PRINT "XHIGH /"XH"/ "": INPUT A$: IF A$ < > "" THEN XH = VAL
(A$):AS% = 0
410 PRINT "YLOW /"YL"/ "": INPUT A$: IF A$ < > "" THEN YL = VAL
(A$):AS% = 0
420 PRINT "YHIGH /"YH"/ "": INPUT A$: IF A$ < > "" THEN YH = VAL
(A$):AS% = 0
422 GOTO 430
424 PRINT "MAG LOW/"ML;"/ "": INPUT A$: IF A$ < > "" THEN ML =
VAL (A$):AS% = 0
425 PRINT "MAG HIGH/"MH;"/ "": INPUT A$: IF A$ < > "" THEN MH =
VAL (A$):AS% = 0
426 PRINT "PHASE LOW/"PL;"/ "": INPUT A$: IF A$ < > "" THEN PL =
VAL (A$):AS% = 0
427 PRINT "PHASE HIGH/"PH;"/ "": INPUT A$: IF A$ < > "" THEN PH
= VAL (A$):AS% = 0
428 PRINT "FREQ LOW/"FL;"/ "": INPUT A$: IF A$ < > "" THEN FL =
VAL (A$):AS% = 0
429 PRINT "FREQ HIGH/"FH;"/ "": INPUT A$: IF A$ < > "" THEN FH =
VAL (A$):AS% = 0
430 REM : INPUT "LABELS (Y/N DEF=N)?":LA$
440 REM : INPUT "TITLE (TOP/BOTTOM/NONE DEF=TOP)?":T$
442 GOTO 80: REM

```

```

445 IF A$ < > "P" THEN 530
447 IF NOT F% THEN PRINT "NO FILE TO PLOT": GOTO 110
450 POKE - 16297,0: POKE - 16304,0: POKE - 16302,0:U = 1
455 IF M% THEN GOSUB 800: GOTO 502: REM PLOT MAGNITUDE/PHASE
457 IF AS% THEN XH = UH:XL = UL:YH = UH:YL = UL: REM PLOT ON UNIT
CIRCLE
460 DX = 279 / (XH - XL):DY = 191 / (YH - YL)
465 SH = 2: REM DEFINE PLOT SHAPE
470 FOR J = 0 TO NP - 1
480 Y = (YH - A(J,1)) * DY + .5:X = (A(J,0) - XL) * DX + .5
490 IF Y > = 0 AND Y < = 191 AND X > = 0 AND X < = 279 THEN HCOLOR=
3: DRAW SH AT X,Y
500 NEXT
502 IF LEFT$ (LA$,1) < > "Y" OR LA$ = "" THEN 512
503 CALL 3072: POKE 792,0: PRINT CHR$ (1): PRINT CHR$ (17): REM
HGR CHR GEN INITIALIZATION

```

```

504 UTAB 24: HTAB 1: PRINT XL:
506 UTAB 24: HTAB 40 - LEN ( STR# (XH)): PRINT XH:
508 UTAB 1: HTAB 40 - LEN ( STR# (YH)): PRINT YH:
510 UTAB 23: HTAB 40 - LEN ( STR# (YL)): PRINT YL: CALL 1013
512 IF LEFT# (T#,1) = "N" THEN 520
515 CALL 3072: POKE 792,0: PRINT CHR# (1): PRINT CHR# (17): REM
      HGR CHR GEN INITIALIZATION
517 UTAB 2 + 22 * (T# = "B"): HTAB 2 + (T# = "B") * (18 - LEN (TI#)
  / 2): PRINT TI#: CALL 1013
520 GOTO 80: REM

530 IF A# < > "H" THEN 560
540 POKE 1145,49: CALL - 16038
550 GOTO 110: REM

560 IF A# < > "T" THEN 620
570 POKE - 16304,0: POKE - 16297,0: U = 1
578 CALL 3072: POKE 792,0: PRINT CHR# (1): PRINT CHR# (17): REM
      HGR CHR GEN INITIALIZATION
579 HTAB 20: UTAB 12
580 GET B#
590 IF ASC (B#) = 13 OR ASC (B#) = 24 THEN 610
600 PRINT B#: GOTO 580
610 CALL 1013: GOTO 80: REM

620 IF A# = "E" THEN TEXT : END : REM EXIT ROUTINE

622 IF A# < > "U" THEN 630: REM TOGGLE TEXT & GRAPHICS SCREENS
624 U = NOT U: IF NOT U THEN POKE - 16303,0: GOTO 110
626 POKE - 16304,0: POKE - 16297,0: POKE - 16302,0
628 GOTO 110: REM

630 PRINT CHR# (7)"ILLEGAL COMMAND"
635 GOTO 110: REM

800 REM ROUTINE TO PLOT MAGNITUDE & PHASE
810 IF AS% THEN FL = FM:FH = FX: REM AUTOSCALE FREQ
815 IF AS% THEN ML = 10 * INT (MM / 10):MH = 10 * INT (MX / 10 +
  1.0): REM AUTOSCALE & ROUND LIMITS TO NEAREST 100B
817 IF AS% THEN PH = 180:PL = - 180: REM AUTOSCALE PHASE AXIS
822 XL = FL:XH = FH: REM SCALE FREQ AXIS
825 GOSUB 950: REM LABEL VERT & HORZ AXIS
828 YL = ML:YH = MH: REM SCALE MAG AXIS
830 DX = 220 / (XH - XL):DY = 170 / (YH - YL)
835 HCOLOR= 3:SH = 2: DRAW SH AT 4,100: REM MAGNITUDE SHAPE
840 FOR J = 0 TO NP - 1: REM PLOT MAGNITUDE
850 X = (A(J,2) - XL) * DX + 29.5:Y = (YH - A(J,3)) * DY + 4.5
855 IF X > 28 AND X < 250 AND Y > 3 AND Y < 175 THEN DRAW SH AT X,Y
860 NEXT J
865 YL = PL:YH = PH: REM SCALE PHASE AXIS
870 DY = 170 / (YH - YL)
875 HCOLOR= 3:SH = 1: DRAW SH AT 270,100: REM PHASE SHAPE
880 FOR J = 0 TO NP - 1: REM PLOT PHASE
890 X = (A(J,2) - XL) * DX + 29.5:Y = (YH - A(J,4)) * DY + 4.5
900 IF X > 28 AND X < 250 AND Y > 3 AND Y < 175 THEN DRAW SH AT X,Y

```



```

910 NEXT J
920 RETURN : REM

```

```

950 REM LABEL VERT & HORZ AXIS
955 CALL 3072: POKE 792,0: PRINT CHR$(1): PRINT CHR$(17): REM
    HGR CHR GEN INITIALIZATION
960 LB = FN R1(FL): UTAB 23: HTAB 3: PRINT LB
965 LB = FN R1(FH): UTAB 23: HTAB 37 - LEN(STR$(XH)): PRINT LB
970 LB = FN R0(MH): UTAB 1: HTAB 1: PRINT LB
975 LB = FN R0(ML): UTAB 22: HTAB 1: PRINT LB
980 LB = FN R0(PH): HTAB 37: UTAB 1: PRINT LB
985 LB = FN R0(PL): HTAB 37:: UTAB 22: PRINT LB
990 CALL 1013: RETURN : REM

```

```

1000 REM ERROR HANDLING
1002 IF PEEK(222) = 254 THEN RESUME
1005 PRINT CHR$(7);"ERROR "; PEEK(222)
1010 POKE - 16303,0
1015 GOTO 110: REM

```

```

***FARPLOT***
TEXT:HOME:INVERSE:HTAB10:PRINT"FARPLOT PROGRAM":NORMAL
BRUN BINARY LIB3.1
PRINT"LIBRARY VERSION ";PEEK(5375)/10
BLOAD HGR CHR GEN
BLOAD SHAPES.A8127
POKE 232,191:POKE 233,31:POKE 230,32
RUN PLOTTER1.A

```